

AIRE VARIABLE

Découverte de la géométrie dynamique avec XCAS

Compétences mathématiques :

- Généralités sur les fonctions.

Compétences informatiques :

- Création de points mobiles dépendant de paramètres formels.

Prérequis :

- Rien de particulier.

Partie 1: Un premier exemple guidé

Le problème

On considère un triangle ABC rectangle en A tel que $AC = 3$ et $AB = 4$. Soit M un point quelconque du segment [AC]. On construit le rectangle AMNP tel que N appartienne à [BC] et P à [AB]. Étudiez l'aire du rectangle AMNP en fonction de la position de M.

1. Construction de la figure

- a. On commence par ouvrir une fenêtre de géométrie en tapant simultanément sur `Alt` et `G` puis définissons les points A et B dans un repère judicieusement choisi. On utilise `point(x,y)` :

```
A:=point(0,0) // on place A
B:=point(0,-4) // on place B tel que AB=4 en laissant le 1er quadrant libre pour y
tracer la courbe à la fin de la séance
```

- b. Définissons ensuite le point C tel que le triangle ABC soit direct, rectangle en A et que l'on ait $AC = \frac{3}{4}AB$ à l'aide de la commande `triangle_rectangle` :

```
triangle_rectangle(A,B,3/4,C)
```

- c. Créons maintenant un réel quelconque de $[0 ; 3]$ que l'on puisse faire varier à la souris à l'aide de la commande `assume(t=[valeur de départ,mini,maxi])` :

```
assume(t=[1,0,3])
```

- d. Définissons maintenant le point M d'abscisse variable t :

```
M:=point(t,0)
```

- e. Pour définir N, on commence par définir la perpendiculaire en M à la droite (AC). La syntaxe est tout à fait naturelle grâce à `perpendiculaire(Point,Droite)`. C'est bien ce que l'élève doit tracer sur sa feuille de papier pour tracer le rectangle :

```
d:=perpendiculaire(M,droite(A,C))
```

- f. Définissons ensuite N comme l'intersection de d et (BC) grâce à `inter_unique` :

```
N:=inter_unique(d,droite(B,C))
```

- g. Pour obtenir P, on définit la parallèle à (AC) passant par N à l'aide de `parallele(Point,Droite)` :

```
D:=parallele(N,droite(A,C));
```

puis l'intersection de D et (AB) :

```
P:=inter_unique(D,droite(A,B))
```

- h. Il ne reste plus qu'à définir le rectangle APNM grâce à la commande `polygone` :

```
R:=polygone(A,P,N,M)
```

et à dessiner le rectangle dépendant du paramètre t :

```
couleur(R,jaune+rempli)
```

- i. En faisant varier t à la souris, le rectangle bouge .

2. Étude de la fonction définie par l'aire du rectangle

a. Définissons maintenant l'expression de l'aire du rectangle dépendant de t . Appelons-la par exemple Surf :

```
Surf:=aire(R)
```

b. Donnons son expression simplifiée :


```
simplifier(Surf)
```

c. Définissons le point X de la courbe représentative de la fonction associée d'abscisse t :

```
X:=point(t, Surf)
```

d. On trace enfin la courbe représentative de la fonction associée à l'aide de la commande graphe :

```
graphe(Surf,t=0..3,couleur=bleu)
```



Partie 2: À vous de jouer

Le problème

On considère un triangle isocèle de sommet principal O avec $OA=6\text{cm}$, $OB=6\text{cm}$. On place M sur $[OA]$ et on note $x = OM$. On place N sur $[OB]$ tel que $BN = OM$. Quand le point M varie sur $[OA]$, le triangle OMN varie. On souhaite étudier l'aire du triangle OMN en fonction de x .

En vous inspirant de ce qui a été fait à la partie précédente, étudiez ce nouveau problème. Vous tracerez la figure à l'aide de Xcas.

Vous y ferez apparaître la courbe représentative de la fonction f qui à x associe l'aire de OMN. Vous demanderez à Xcas l'expression de cette aire en fonction de x .